
A bare-bones setup
In spite of my occasional involvement in the 4TEX project
and the existence of a 4TEX installation at our department,
I am not a 4TEX user. Ever since I started using TEX at
the end of 1991, I have used my own configuration files in
combination with the existing 4TEX installation: I wanted
to be able to run the compile–preview–print cycle, prefer-
ably from the commandline, with only a minimum of re-
configuration of my system.

If that is also what you want, you may want to look at
the batchfiles in the \rawemtex subdirectory of the 4TEX
workbench cd. However, these only cover emtex itself.

Currently I use a single batchfiletexcfg.bat of my own
which does all the configuring. It defines some environ-
ment variables and DOSKEY macros, making use of the
pre-existing configuration files. After it has run, the core
components of TEX can be run with a simple command–
and–filename commandline.

It does nothing particularly clever and doesn’t offer the
functionality of 4TEX, but it does – contrary to the
rawemtex batchfiles – also integrate PostScript support
and several utilities. Adopting it may help you in the task
of fitting all the pieces together.

The CD version is available at www.eco.rug.nl/
medewerk/siep/texcfgcd.zip. The new edition
of the 4allTEX CD will contain a matching version. Several
colleagues in my department use a similar batchfile adapted
to the TEX-installation on our LAN.

How far do you get?
Supported are the compiler, emtex and PostScript dvi-
drivers with support for Metafont automatic font genera-
tion, BibTEX, MakeIndex and Ghostscript. Essential op-
tions are taken care of, but additional options will be passed
on to the respective programs.

Block-compile is not explicitly supported but requires only
a little bit of manual setup: create a small auxiliary TEX
document frame.tex consisting of
1. the preamble of the document you are working on, up

to and including the line begin{document}
2. a line \input tmp.tex

3. a line \end{document}

When you want to compile just a fragment of your docu-
ment, save that fragment as tmp.tex. You can write a
macro for this. Then compile frame.tex. If you work
under Windows you might want to edit and compile in sep-
arate windows so that your document can remain loaded in
your editor during compilation.

The most important 4TEX features which are not supported
and are non-trivial to set up are automatic font generation
for Type 1 fonts and graphics conversion facilities. You
can use Type 1 fonts and graphics with the texcfg configu-
ration, if you use the PostScript dvi-driver dvips and don’t
mind previewing and maybe also printing with Ghostscript.
Dvips can use Type 1 fonts directly and can include graph-
ics in eps- or Encapsulated Postscript format. Since vir-
tually any graphics program can export directly in eps-
format, and eps is even for bitmaps often the preferred for-
mat, this is no limitation.

Ghostscript runs quite well on current PC hardware, but
you do sacrifice some convenience with previewing and
some quality with printing if you don’t have a PostScript
printer.

Installation and customization
Copy texcfg.ba_ to a file texcfg.bat on your hard-
disk, editing it as necessary. See the comments in the file
itself for details. As noted there, you also need a suf-
ficiently large environment and for doskey a sufficiently
large buffer. Finally, as is also indicated there, you need
to place a copy of Ghostscript’s dos extender dos4gw.exe
on your path.

I have tried to make basic customization easy: there are six
clearly marked places where changes may be necessary to
get it working, and many more where customization is op-
tional. If you have a PCL- or PostScript printer there prob-
ably is already a DOSKEY macro which you can activate
by uncommenting.

Additional customization may be desirable if you want to
use fonts or style files of your own, and also if you don’t
keep all sources for one document in one directory, as is the
assumption of texcfg. Texcfg contains ample comments to
guide you in such customization.

It is fairly trivial to replace the doskey macros with 4dos
aliases or with regular batchfiles. You might want the lat-
ter for fully automated TEX runs since doskey macros can’t
be called from batchfiles.

1 Ghostscript under Windows
If you want to use Ghostscript’s Windows executable,
gswin or gswin32, you can add an icon for it in the usual
way with Windows’ Program Manager. You can find
it in the same directory as the DOS executable, i.e. in
\emtex\utils\ghostscr. It doesn’t require any ad-
ditional configuration beyond what is already done.

In the same directory you will find gsview[32], which adds
a proper Windows interface to gswin. When you start it for

the first time it will tell you how to configure it. Be fore-
warned though that gswin runs much better by itself.

2 Dvips
I couldn’t entirely avoid custom configuration files since
the dvips Metafont mode can only be set in a configuration
file. Texcfg creates at startup a printer-specific or rather
resolution-specific configuration file%printer%.cfg in
the root directory of the c:-drive with the proper Metafont
mode.

Since this configuration file had to be created anyhow, I
also added a line p +MYPSMAP.MAP to facilitate the use
of additional Type 1 fonts. You can safely ignore this fea-
ture if it means nothing to you.

@echo off
:: TeX from the command-line version for 4allTeX 3rd edition CD1:
:: 4TeX workbench 20 april 1996

:: Create or erase TeX configuration:
:: EMTEXDIR undefined or 0: create
:: EMTEXDIR defined and <>0: erase

:: !!!! Check AT LEAST the settings for !!!!!!!!!!!!!!!!!!!!!!!!!!!
:: - EMTEXDIR (drive letter cd-rom)
:: - initial loading of doskey with enlarged buffer
:: - LATEX (209: old or 2E: new)
:: - PRINTER (printer resolution for POSTSCRIPT printing;
:: laserjet for 300dpi, ljIV for 600dpi)
:: irrelevant for PCL printing
:: - doskey tprint (PCL printer choice; print to file or to printer)
:: - doskey gprint (printer choice)
:: !!!! These items are marked like this !!!!!!!!!!!!!!!!!!!!!!!!!!
:: This batchfile might place a file ljiv.cfg in c:\
:: Limitations
:: -----------
:: - 86- en -286 versions not supported
:: - Automatic font generation supported only for MetaFont fonts
:: - memory management: is o.k. if either TeX is run from a Windows DOS
:: box, or in CONFIG.SYS [Q]EMM386 is called with a RAM parameter
:: (for expanded memory). In the first case RSX is used, in the second EMX.

:: !!!! BOOT FILES !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
:: CONFIG.SYS:
:: The SHELL command should specify a sufficiently large environment, eg.
:: SHELL=C:\COMMAND.COM C:\ /E:1000 /P
:: AUTOEXEC.BAT:
:: If doskey is already loaded there, it should specify a sufficiently
:: large buffer, eg.
:: lh doskey /bufsize=1000
:: Also, copy EMTEX\UTILS\GHOSTSCR\DOS4GW.EXE to a directory on your path.

:: BEGINNING OF CODE ::::::::::::::::::::::::::::::::::::::::::::::
SET OPT=
IF %EMTEXDIR%[==[ SET OPT=INSTALL
IF %EMTEXDIR%[==0[ SET OPT=REINSTALL
IF %OPT%[==[ GOTO ERASE

:INSTALL
:: !!!! drive letter cd-rom !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
SET EMTEXDIR=d:\EMTEX
SET EMTEXDIRSLASH=d:\EMTEX\

IF %OPT%==REINSTALL GOTO PATHDONE

:: !!!! Uncomment if doskey not loaded previously !!!!!!!!!!!!!!!!!
:: lh doskey /bufsize=1000

:: Addition to search path
PATH %PATH%;%EMTEXDIRSLASH%METAFONT

:PATHDONE
SET OPT=

:: ENVIRONMENT VARIABLES ::::::::::::::::::::::::::::::::::::::::::
:: memory manager: RSX requires Windows DOS box; EMX requires EMM
SET RSX=%EMTEXDIRSLASH%RSX.EXE
SET EMX=%EMTEXDIRSLASH%EMX.EXE

:: !!!! old or new LaTeX !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
:: SET LATEX=209
SET LATEX=2E

:: TeX input files (.STY); may add own directories
IF %LATEX%!==2E! SET TEXINPUT=%EMTEXDIRSLASH%LATEX!;%EMTEXDIRSLASH%INPUTS!;%EMTEXDIRSLASH%LATEX209!
IF %LATEX%!==209! SET TEXINPUT=%EMTEXDIRSLASH%LATEX209!;%EMTEXDIRSLASH%INPUTS!;%EMTEXDIRSLASH%LATEX!

:: bibtex input files

SET BIBINPUT=.

:: precompiled TeX formats
SET BTEXFMT=%EMTEXDIRSLASH%BTEXFMTS

:: TeX font metrics; may add own directories
SET TEXTFM=.;%EMTEXDIRSLASH%TFM!
:: same for dvips:
SET TEXFONTS=%TEXTFM%

:: !!!! PostScript metafont mode, or printer resolution: !!!!
:: laserjet=300dpi; ljIV=600dpi
:: There should be a file \emtex\ps\%printer%.cfg and the Metafont
:: mode should be listed in \emtex\metafont\mfjob\modes.mfj
:: irrelevant for non-PostScript printing
SET PRINTER=laserjet
:: SET PRINTER=ljIV

:: MetaFont files (.MF)
SET MFINPUT=%EMTEXDIRSLASH%metafont\mfinput!

:: MetaFont base files
SET BMFBAS=%EMTEXDIRSLASH%METAFONT\BMFBASES

:: Metafont job files
SET MFJOB=%EMTEXDIRSLASH%METAFONT\MFJOB

:: MFJOB options
SET MFJOBOPT=/i /3

:: base directory for generated fonts
SET DVIDRVFONTS=C:\TEXFONTS

:: same for DVIPS:
SET TEXPKS=%DVIDRVFONTS%\%%ddpi\%%f.%%p;%EMTEXDIRSLASH%TEXFONTS\%printer%\%%ddpi\%%f.%%p

:: Virtual fonts (for dvips)
SET VFFONTS=.;%EMTEXDIRSLASH%TEXFONTS\VF

:: directory for own font library ownfli.fli
:: the dvidrv cnf files use the environment variabele OWNFLI,
:: the dvips configuration file config.ps uses MYFONTS.
SET OWNFLI=%DVIDRVFONTS%
SET MYFONTS=%OWNFLI%

:: a few additional bells and whistles of the emTeX dvi drivers;
:: these environment variables may go undefined if dvi files
:: and graphic files are always in the current directory.
:: SET DRVDRVINPUT=.
:: SET DVIDRVGRAPH=.

:: some default options for the emtex dvi drivers
:: /fm: Metafont mode (see emtex\metafont\mfjob\modes.mfj)
:: These settings can be overruled on the dvidrv command-line.
SET DVIDRV=/fb /ft:1 /fj+ /fm:laserjet
SET DVIHPLJ=/po:lpt1
SET DVIDOT=/po:lpt1 /fm:epsonfx

:: DVIPS directory for configuration files; will add ‘C:\’ later
SET TEXCONFIG=%EMTEXDIRSLASH%PS

:: dvips printer-specific configuration file ::::::::::::::::::::::
:: If you want to use your own PostScript fonts,
:: you can make them known to dvips in C:\MYPSFONTS.MAP

COPY %TEXCONFIG%\%PRINTER%.CFG C:\
ECHO M %PRINTER% >>C:\%PRINTER%.CFG
ECHO p +MYPSMAP.MAP >>C:\%PRINTER%.CFG

SET TEXCONFIG=C:\;%TEXCONFIG%

:: dvips header files, eg. fonts; may add own directory
SET DVIPSHEADERS=%EMTEXDIRSLASH%PS;%EMTEXDIRSLASH%PS\FONTS

:: Where GhostScript looks for files (apart from current directory).
:: This concerns initialization files, fonts and the file fontmap.
:: ***
SET GS_LIB=%EMTEXDIRSLASH%PS\FONTS;%EMTEXDIRSLASH%UTILS\GHOSTSCR;%EMTEXDIRSLASH%UTILS\SOURCE

:: DOSKEY MACRO’S :::::::::::::::::::::::::::::::::::::::::::::::::

:: NOTE. doskey syntax: ‘$*’ means ‘%1 ... %9’

:::: compiling ::::

IF %LATEX%!==2E! doskey latex=%EMTEXDIRSLASH%compiler\tex386 -mt28000 &latex2e $*
IF %LATEX%!==209! doskey latex=%EMTEXDIRSLASH%compiler\tex386 -mt28000 &lplainn $*
IF %LATEX%!==209! doskey slitex=%EMTEXDIRSLASH%compiler\tex386 -mt28000 &splainn $*

:::: emTeX dvi drivers: previewing and printing ::::
:: Note. ‘@hplj.cnf’ calls a ‘response file’ hplj.cnf with options.
:: See emtex\data for additional configuration options.

:: previewing; automatic adapter selection
:: Windows users: be sure to also check out emtex\mswindow\dviwin
doskey tview=%EMTEXDIRSLASH%dvidrv %EMTEXDIRSLASH%dviscr /or+ /s3 @hplj.cnf @a4.pap $*

:: !!!! choose a tprint version or define your own !!!!!!!!!!!!!!!!
:: You also may want to have a look at the batchfiles in \rawemtex
:: LaserJet <=3 and printing to <filename>.pcl
:: doskey tprint=%EMTEXDIRSLASH%dvidrv %EMTEXDIRSLASH%dvihplj @hplj.cnf /po=@Bi.pcl @a4.pap $*

:: LaserJet >=4 and printing to <filename>.pcl
doskey tprint=%EMTEXDIRSLASH%dvidrv %EMTEXDIRSLASH%dvihplj @ljiv.cnf /fm:ljIV /og600 /po=@Bi.pcl @a4.pap $*

:: DeskJet and printing to printer (use LaserJet driver)
:: doskey tprint=%EMTEXDIRSLASH%dvidrv %EMTEXDIRSLASH%dvihplj @hplj.cnf /op:deskjet @a4.pap $*

:: IBM Proprinter 4201: a non-pcl, non-PostScript printer using dvidot
:: doskey tprint=%EMTEXDIRSLASH%dvidrv %EMTEXDIRSLASH%dvidot ibm4201 @fx $*

:::: PostScript dvi driver ::::
:: Note. dvips writes default to <filename>.ps
:: A configuration file %printer%.cfg is read automatically.
doskey tps=%EMTEXDIRSLASH%dvidrv %EMTEXDIRSLASH%dvips32 $*

:::: GhostScript: PostScript for non-PostScript devices ::::
:: ‘gs386 -?’ shows available devices and the more important options

:: Viewing PostScript files. May replace vga with eg. s3vga or tseng.
:: Windows users: gswin.exe with or without gsview.exe is preferable
:: but should be installed from the Windows Program Manager.
doskey gview=%EMTEXDIRSLASH%UTILS\GHOSTSCR\gs386 -q -sDEVICE=vga $* quit.ps

:: Printing PostScript files to a non-PostScript printer

:: !!!! Replace ljet3 with eg. deskjet or djet500? !!!!!!!!!!!!!!!!
doskey gprint=%EMTEXDIRSLASH%UTILS\GHOSTSCR\gs386 -q -sDEVICE=ljet3 $* quit.ps

:::: bibtex ::::
doskey tbib=%EMTEXDIRSLASH%UTILS\bibtex32 $*

ECHO TeX configured
goto end

:: ERASING ::::::::::::::::::::::::::::::::::::::::::::::::::::::::
:ERASE
SET EMTEXDIR=0
SET EMTEXDIRSLASH=
SET RSX=
SET EMX=
SET LATEX=
SET TEXINPUT=
SET BIBINPUT=
SET BTEXFMT=
SET TEXTFM=
SET TEXFONTS=
SET PRINTER=
SET MFINPUT=
SET BMFBAS=
SET MFJOB=
SET MFJOBOPT=
SET DVIDRVFONTS=
SET TEXPKS=
SET VFFONTS=
SET OWNFLI=
SET MYFONTS=
SET DRVDRVINPUT=
SET DVIDRVGRAPH=
SET DVIDRV=
SET DVIHPLJ=
SET TEXCONFIG=
SET DVIPSHEADERS=
SET GS_LIB=

IF EXIST C:\LJIV.CFG DEL C:\LJIV.CFG

DOSKEY LATEX=
DOSKEY SLITEX=
DOSKEY TEX=
DOSKEY TVIEW=
DOSKEY TPRINT=
DOSKEY TPS=
DOSKEY GVIEW=
DOSKEY GPRINT=
DOSKEY TBIB=

echo TeX configuration erased
:end


